



# DECUS

## PROGRAM LIBRARY

DECUS NO.	8-694
TITLE	TELETYPE LINE PRINTER EMULATOR HANDLER FOR OS/8
AUTHOR	Stanley R. Vivian
COMPANY	University of Manitoba Winnipeg, Canada
DATE	March 25, 1974
SOURCE LANGUAGE	PAL-8

### ATTENTION

This is a USER program. Other than requiring that it conform to submittal and review standards, no quality control has been imposed upon this program by DECUS.

The DECUS Program Library is a clearing house only; it does not generate or test programs. No warranty, express or implied, is made by the contributor, Digital Equipment Computer Users Society or Digital Equipment Corporation as to the accuracy or functioning of the program or related material, and no responsibility is assumed by these parties in connection therewith.

TELETYPE LINE PRINTER EMULATOR HANDLER FOR OS/8

DECUS Program Library Write-up

DECUS NO. 8-694

ABSTRACT

Teletype Line Printer Emulator Handler for OS/8

Stanley R. Vivian, University of Manitoba, Winnipeg, Canada.

This OS/8 handler emulates the LPO8 line printer on the ASR33 teletype. It handles form-feeds, tabs, line overrun and paging. A character count automatically generates a carriage return-line feed whenever the count exceeds 72. A line count automatically pages at 62 lines by introducing 4 additional CR/LF's to produce 11 inch pages. Due to space limitations in the handler, vertical tab results in a single additional CR/LF. An attempt to read from the handler results in an immediate normal exit. First entry to the handler generates 4 CR/LF's. There is no closing form-feed.

Handler modified for use with  
LPT: via KL87A card with device  
code = 15.

## DESCRIPTION

This is a replacement LPT: handler for OS/8 for those of us unfortunate enough not to have a line printer but who want the formatting capability of the LPO8 handler. The handler is set up for the standard ASR33 friction feed teletype but could be used with other printers as is, or adapted to them if necessary.

The handler generates 4 line feeds on first entry and begins paging from that point - this is analogous to the opening form-feed of the LPO8 handler. Tabs result in spacing to the next 8th print position as with the LPO8 or the /T option of PIP or the EDITOR. A character count is maintained and a CR/LF generated whenever the character count reaches 72. The extra line feed goes into the line count preserving paging. The page count automatically introduces 4 extra LF/CR's whenever it reaches 62, to give 11 inch pages. This feature differs from the LPO8 handler in that long FOCAL, FORTRAN or BASIC programs will now be paged. A form-feed encountered anywhere in the buffer will result in sufficient CR/LF's to reach the next "top of page", that is, relative to the first entry. PAL8 or CREF listings normally page at 57 lines by introducing a form-feed into the buffer. This handler will accomodate several line overruns without overflowing onto a new page for such listings. If many line overruns are anticipated, as in heavily commented CREF listings, it would be advisable to generate non-paginated output with PAL8 via the /H option. This handler will then provide the necessary paging.

Due to space limitations in the handler, it was not possible to generate the customary 9 line-feeds when encountering vertical tab. A single line-feed is generated. This could be changed to generate a form-feed if desired.

The only possible error, an attempt to read from the handler, results in an immediate normal exit - there was not enough room in the handler to generate a negative AC for a fatal exit.

## MODIFYING THE MONITOR

1. Create a binary file

```
.R PIP
*OS8LTY.PA<PTR:

.R PAL8
*OS8LTY,<OS8LTY
```
2. Incorporate the handler into OS/8

```
.R BUILD
$LOAD DSK:OS8LTY
$INSERT LTY,LPT
$BOOT
.SAVE SYS BUILD 0000-7577,10000-17577;200=0000
```

The user may have to delete some non needed active devices, such as the upper DECTape drives, if there is insufficient room for another device. Users with TC01/TC08 systems as the system device should also declare their system SYS=TC08 if they want the default device to be other than DTAO.

#### OPERATION INSTRUCTIONS

The user simply sends output to LPT that he would have previously sent to TTY. The accompanying listing was produced as follows:

```
.R PAL8  
*,LPT:<OS8LTY
```

/TTY: EMULATOR FOR LINE PRINTER

/EMULATES THE LP08 LINE PRINTER ON THE ASR33. HANDLES  
 /TABS, LINE OVERFLOW (72 CHARACTERS), FORM FEEDS, AND  
 /AUTOMATIC PAGING. VERTICAL TAB GENERATES A SINGLE CR/LF.  
 /THERE ARE NO ERRORS - AN ATTEMPT TO READ FROM THE LPT:  
 /RESULTS IN AN IMMEDIATE EXIT WITHOUT ERROR. AUTOMATIC  
 /PAGING IS AT 62 LINES.

/HEADER BLOCK

```

00000 0000 *0
00000 7777 -1 /ONLY 1 DEVICE
00001 1424 DEVICE LTY; DEVICE LPT; 1040; 0000; ZBLOCK 2
00002 3100
00003 1420
00004 2400
00005 1040
00006 0000
00007 0000
  
```

0200 \*200

6041 TSF=6041  
 6046 TLS=6046

```

00200 0000 LPT, 0 /ENTRY HERE
00201 7330 CLL CML CLA RAR /SET AC TO 4000, LINK TO 0
00202 1600 TAD I LPT /LINK GOES ON HERE. IF NOT READ WAS
REQUESTED
00203 0215 LPT215, AND LP7700 /MASK FUNCTION WORD TO GET
00204 7040 CMA / WORD COUNT
00205 3262 DCA LPTWC
00206 6214 LPT214, RDF /SET UP RETURN
00207 1317 TAD LPTCIF / BY FORMULATING A 'CDI'
00210 3260 DCA LPTXIT /SAVE CALLING FIELD IN RETURN SEQUE
NCE
00211 1600 TAD I LPT /RECALL THE FUNCTION WORD TO GET
00212 0355 AND LPT70 / DATA FIELD BITS
00213 1362 TAD LPCDF0 /FORMULATE THE DATA FIELD
00214 3223 DCA LPTCDF
  
```

/NOTE!!!!!!!

```

00215 7700 LP7700, 7700 /MUST BE AT LOC. 15 OF PAGE!
00216 2200 ISZ LPT /GET NEXT ARG.
00217 1600 TAD I LPT /STARTING ADDRESS OF DATA
00220 3263 DCA LPTCA
00221 2200 ISZ LPT /NEXT ARG.
00222 1600 TAD I LPT /BLOCK NO. - GENERATE FF FIRST ENTR
Y
00223 7402 LPTCDF, HLT /'CDF' OVERLAID HERE
00224 2200 ISZ LPT /MOVE PAST LAST ARG.
00225 7420 SNL /IF LINK NOT SET, READ REQUESTED
00226 5257 JMP LPRTRN /READ WANTED - EXIT NOW
00227 7650 SNA CLA /IS THIS FIRST ENTRY?
00230 1206 TAD LPT214 /YES - LOAD A FF TO INITIALIZE
00231 5253 JMP LPTKLG /BY-PASS UNPACKING AND JUMP IN
  
```

/NOW PICK UP THE CHARACTERS, UNPACK THEM AND PRINT.  
 /TEST FOR:TABS, LINE FEEDS,VERTICAL TABS, FORM FEEDS.

00232	1663	LPTLP,	TAD I LPTCA	/GET FIRST CHAR. OF 1-2-3
00233	4264		JMS LPTPCH	/PRINT IT OR WHATEVER
00234	1663		TAD I LPTCA	/GET LEFT 4 BITS OF 3RD CHAR.
00235	0240		AND LP7400	
00236	3223		DCA LPTCDF	/LOCATION NOW AVAILABLE FOR TEMP.
00237	2263		ISZ LPTCA	/NEXT WORD
00240	7400	LP7400,	7400	/EFFECTIVE NOP WHEN EXECUTED
00241	1663		TAD I LPTCA	/GET 2ND CHAR.
00242	4264		JMS LPTPCH	/DO ITS THING
00243	1663		TAD I LPTCA	/NOW GET REST OF 3RD CHAR.
00244	0240		AND LP7400	
00245	7112		CLL RTR	/ASSEMBLE THE TWO 4 BITS
00246	7012	LPTLF,	RTR	
00247	1223		TAD LPTCDF	
00250	7012		RTR	
00251	7012		RTR	
00252	2263		ISZ LPTCA	/NEXT WORD
00253	4264	LPTKLG,	JMS LPTPCH	/LET IT DO ITS THING
00254	0177	LPT177,	177	/EFFECTIVE NOP AS AC IS CLEAR
00255	2262		ISZ LPTWC	/ANY MORE DATA?
00256	5232		JMP LPTLP	/YES - GO FOR MORE
00257	2200	LPRTN,	ISZ LPT	/NO - MOVE PAST ERROR RETURN
00260	7402	LPTXIT,	HLT	/OVERLAID WITH THE PROPER 'CDI'
00261	5600		JMP I LPT	/HOME
00262	0000	LPTWC,	0	/WORD COUNT
00263	0000	LPTCA,	0	/CURRENT DATA ADDRESS
00264	0000	LPTPCH,	0	/SUB. TO SCREEN FOR SPECIAL CHAR.
00265	0254		AND LPT177	/KEEP ONLY THE GOOD THINGS
00266	7041		CIA	/SCREENS OUT 'ESCAPE' & 'RUBOUT'
00267	7120		CLL CML	
00270	1360		TAD LPT175	
00271	7470		SZL SNA	/JUMP OUT ON CODES 176-177
00272	5354		JMP LPM140	
00273	1361		TAD LPM143	/CHECK FOR CTRL/Z
00274	7450		SNA	/GOT IT?
00275	5257		JMP LPRTN	/YES - EXIT (SORRY NO ROOM FOR
00276	1357		TAD LPTM15	/ CLOSING FF)
00277	7160		CLL CML CMA	/SET UP TO TRAP CODES 211-216
00300	1347		TAD LPT7	/ AND JMP OUT WITH A CONSTANT
00301	7460		SNL SZA	/ IN THE AC
00302	5336		JMP LPTCTL	/ AC=1 FOR TAB, 2 FOR LF, ETC.
00303	1347		TAD LPT7	/RESTORE THE INITIAL CODE
00304	4306		JMS LPWAIT	/PRINT THE CHAR.
00305	5346	LPTJMP,	JMP LPJTAB-2	/LPJTAB-2 = "JMP I LPTPCH"
00306	0000	LPWAIT,	0	/PRINT ROUTINE
00307	6046		TLS	
00310	7600	LP7600,	7600	/DOES A CLA
00311	1310		TAD LP7600	/CHECK KB FOR CTRL/C
00312	6034		KRS	

00313	1360		TAD LPT175	
00314	7650		SNA CLA	/GOT IT?
00315	6031		KSF	/YES - EXIT TO MONITOR
00316	5321		JMP .+3	/NO - GO ON
00317	6203	LPTCIF,	CDF CIF 0	/OUT TO FIELD 0 - 7600
00320	5710		JMP I LP7600	
00321	6041		TSF	/WAIT FOR PRINTER TO FINISH
00322	5310		JMP LP7600	/KEEP CHECKING WHILE YOU WAIT
00323	2363	LPWISZ,	ISZ LPLPTR	/IS LINE FULL?
00324	5706		JMP I LPWAIT	/NO - GO FOR MORE
00325	1364	LPLF,	TAD LPM112	/RESET CHAR. COUNT TO 74 (CR/LF INC
		L.)		
00326	3363		DCA LPLPTR	
00327	1203		TAD LPT215	/LP08 DOES A CR/LF ON LF - IGNORES
		CR		
00330	4306		JMS LPWAIT	
00331	1246		TAD LPTLF	
00332	4306		JMS LPWAIT	
00333	2375		ISZ LPTLC	/62 LINES YET?
00334	7410		SKP	/NO - FALL THRU LPTCTL & RETURN
00335	5365		JMP LPTFF	/YES - FEED A FORM
00336	1305	LPTCTL,	TAD LPTJMP	
00337	3340		DCA .+1	/SETS UP FOR SPECIAL CHAR.
00340	7402	LPTXXX,	HLT	/ ROUTINE JMP
00341	1354	LPTTAB,	TAD LPM140	/HORIZONTAL TABS
00342	4306		JMS LPWAIT	
00343	1363		TAD LPLPTR	
00344	0347		AND LPT7	
00345	7650		SNA CLA	
00346	5664		JMP I LPTPCH	/MUST BE AT LPJTAB-2!!
00347	0007	LPT7,	7	
00350	5341	LPJTAB,	JMP LPTTAB	
00351	5325		JMP LPLF	
00352	5325		JMP LPLF	/SORRY - NO ROOM FOR USUAL 9 LF'S
00353	5365		JMP LPTFF	
00354	7640	LPM140,	-140	/CLA WHEN EXECUTED (7640)
00355	0070	LPT70,	70	/NOP WHEN EXEC.
00356	5664		JMP I LPTPCH	/END OF CHARACTER HANDLING ROUT.
00357	7763	LPTM15,	-15	
00360	0175	LPT175,	175	
00361	7635	LPM143,	-143	
00362	6201	LPCDF0,	CDF 0	
00363	0000	LPLPTR,	0	/LINE CHARACTER COUNT
00364	7666	LPM112,	-112	/74 CHAR. LINES - INCLUDES CR/LF
00365	1375	LPTFF,	TAD LPTLC	/MAKE UP TOTAL OF 62+3+1 LINES
00366	1376		TAD M3	/ FOR 11 INCHES
00367	3363		DCA LPLPTR	/LOAD CHAR. COUNT WITH REMAINING

00370	1377		TAD M77	/RE-INITIALIZE THE LINE COUNT
00371	3375		DCA LPTLC	
00372	1246		TAD LPTLF	/SET UP FOR REPEATED LF'S
00373	4306		JMS LPWAIT	/WAIT FOR CHAR. COUNT TO OVERFLOW
00374	5372		JMP .-2	/ AND TAKE OVERFLOW EXIT
				/ WHICH FEEDS 1 MORE LINE!
00375	0000	LPTLC,	0	/LINE COUNT - FIRST FF ADVANCES
				/ ONLY 4 LINES
00376	7775	M3,	-3	
00377	7701	M77,	-77	/LINE COUNT TRIPS AT 62
				/ INITIALIZED AT 63 AS FINAL
				/ LF OF FF DROPS COUNT

SSSSS



LPCDF0 0362  
LPJTAB 0350  
LPLF 0325  
LPLPTR 0363  
LPM112 0364  
LPM140 0354  
LPM143 0361  
LPRTRN 0257  
LPT 0200  
LPTCA 0263  
LPTCDF 0223  
LPTCIF 0317  
LPTCTL 0336  
LPTFF 0365  
LPTJMP 0305  
LPTKLG 0253  
LPTLC 0375  
LPTLF 0246  
LPTLP 0232  
LPTM15 0357  
LPTPCH 0264  
LPTTAB 0341  
LPTWC 0262  
LPTXIT 0260  
LPTXXX 0340  
LPT175 0360  
LPT177 0254  
LPT214 0206  
LPT215 0203  
LPT7 0347  
LPT70 0355  
LPWAIT 0306  
LPWISZ 0323  
LP7400 0240  
LP7600 0310  
LP7700 0215  
M3 0376  
M77 0377

